Customizing PROC FREQ’s Output for One- and Two-Way Tables
Nan Liu, Adheris Inc., Burlington, MA
Marcelo Coca-Perraillon, Adheris Inc., Burlington, MA

ABSTRACT
The PROC FREQ is commonly used to display one- and two-way frequency tables. By default, frequencies are displayed without commas and percentages have two decimal places and no percent sign. One-way tables can be customized by modifying the default Output Delivery System (ODS) template. Unfortunately, there is little customization possible for two-way tables not displayed as a list. In this paper we describe how to customize the PROC FREQ’s output by combining the ODS system with PROC FORMAT and PROC TEMPLATE. The final outputs are one- and two-way tables that can display both the frequencies and the percents with any type of format, including special characters and percent signs.

INTRODUCTION
The PROC FREQ is perhaps one of the most commonly used SAS® procedures. The power of PROC FREQ lies in its wide range of statistical tests, from a simple chi square test to exact tests and concordance measures. However, the display of the tabulations cannot be customized without having to work with SAS templates. SAS offers other, more powerful procedures to display customized reports, like PROC TABULATE and PROC REPORTS, but it is often not convenient to use two different procedures to display and analyze data. This paper presents a way to change the display of tables by working with SAS templates. It also serves as an introduction to working with SAS templates, which allow users to change the display of SAS output.

DATA
The following code creates a sample dataset called SampleData. These data are used throughout the paper to illustrate the output of PROC FREQ. The dataset contains only two variables, annual income and an indicator of college education, for eight individuals:

```
* Create sample data;
data SampleData;
  input College Income;
datalines;
  1 40000
  1 40000
  1 50000
  1 50000
  0 40000
  0 40000
  0 50000
;
```

ONE-WAY TABLES
This section explains how to locate the template used by PROC FREQ and how to change the header and the display of the relative frequencies. The objective is to change the headers “Percent” to “Relative Frequency” and “Cumulative Percent” to “Relative Cumulative Percent.” In addition, we add a percent sing to the relative frequencies.
SAS TEMPLATES

The ODS trace statement locates the template used in any procedure. For instance, the following code finds the template used by PROC FREQ:

```sas
* Find the template;
ods trace on;
proc freq data= SampleData;
tables income;
format income dollar10.0;
run;
ods trace off;
```

The combined log and list output are displayed in Figure 1. Note that the format statement assigns a format to the values of the variable income, but it cannot be used to change the format of the frequencies or percents.

![Figure 1: Combined Log and List Output](image)

<table>
<thead>
<tr>
<th>Income</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Cumulative Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>$40,000</td>
<td>5</td>
<td>62.50</td>
<td>5</td>
<td>62.50</td>
</tr>
<tr>
<td>$50,000</td>
<td>3</td>
<td>37.50</td>
<td>8</td>
<td>100.00</td>
</tr>
</tbody>
</table>

The output of the trace statement is written to the log (Figure 1) and it indicates that the name of the template is Base.Freq.OneWayFreqs and that the template is located under the path Freq.Table1.OneWayFreqs. The next step is to figure out how the template is defined. The source code can be retrieved by submitting the following code:

```sas
* Template source code;
proc template;
source Base.Freq.OneWayFreqs;
run;
```

The source code is reproduced in Appendix A. The sections of the template that control the appearance of the percent and cumulative percent columns are highlighted and they indicate that the display of both columns is actually defined by the parent template Base.Freq.Percent. This is where working with SAS templates can get complicated. The display of an element can be controlled by another template. In our example, the template for Base.Freq.OneWayFreqs defines Base.Freq.Percent as the “parent” of both Percent and CumPercent. In order to change the headers and display the percent sign, we need to modify the template Base.Freq.Percent. To see how Base.Freq.Percent is defined, you can use PROC TEMPLATE again:

```sas
* Template source code;
proc template;
source Base.Freq.Percent;
run;
```

The definition is shown in Figure 2. Note that the definition includes the header, format and justification of the title.
A template can be edited with PROC TEMPLATE. You can either make a permanent change to the template definition or a temporary one for the duration of the SAS session (it is probably a good idea to make changes temporary). To make a temporary change, you can store the modified template into work.template (instead of sashelp.tmplmst) by either declaring the ODS path before PROC TEMPLATE or using the store statement inside PROC TEMPLATE. For our example, we temporarily modify the template using ODS:

```sas
ods path work.templat(update) sashelp.tmplmst(read);
```

The next step is to create a picture format to display the percent sign:

```sas
* Create picture format for percents;
proc format;
picture pctfmt low-high='000.00%';
run;
```

PROC FORMAT’s picture statement allows users to add punctuation to the display of numbers. With the picture statement, you can add plus or minus signs, format phone numbers or add currency symbols. In our example, we define the picture format to display numbers with two decimals and a percent sign at the end.

Finally, we use PROC TEMPLATE with the statement edit to update the Base.Freq.OneWayFreqs template:

```sas
* Edit the template using PROC TEMPLATE;
proc template;
edit Base.Freq.OneWayFreqs;
edit Percent;
  header="; Relative; Frequency ";
  format= pctfmt.;
  justify= on;
end;
edit CumPercent;
  header = ";Cumulative; Relative Frequency ";
  format= pctfmt.;
  justify= on;
end;
end;
run;
```

Note that there are three edit/end blocks. The outermost edit/end block is used to indicate that the Base.Freq.OneWayFreq template is to be updated. The nested blocks are used to edit the definitions of Percent and CumPercent inside Base.Freq.OneWayFreq, which are highlighted in Appendix A. In essence, we are asking SAS to rewrite the source code, changing the headers and formats. However, we are also adding another line by using the column attribute “justify” in order to center the format field within the column. The semicolon in the header “; Relative; Frequency” is used to add a line break.

To check out modified output, we submit PROC FREQ again:

```sas
* Check the new output;
proc freq data= SampleData;
tables income;
  format income dollar10.0;
run;
```

The output is displayed in Figure 3.
TWO-WAY TABLES

Unfortunately, there is no template available to change the output of two-way tables not displayed as lists. To see a two-way table displayed as list, you can use the option “list” in PROC FREQ:

```latex
ods trace on;
proc freq data= SampleData;
tables income * college / list;
format income dollar10.0;
run;
ods trace off;
```

The combined log and list output is displayed in Figure 4. The template that controls the display of the list is called Base.Freq.List (Figure 4). The source code of Base.Freq.List is in Appendix B. The code is similar to that of Base.Freq.OneWayFreqs.

Figure 3
Modified Output

<table>
<thead>
<tr>
<th>Income</th>
<th>Frequency</th>
<th>Relative Frequency</th>
<th>Cumulative Frequency</th>
<th>Cumulative Relative Frequency</th>
</tr>
</thead>
<tbody>
<tr>
<td>$40,000</td>
<td>5</td>
<td>62.50%</td>
<td>5</td>
<td>62.50%</td>
</tr>
<tr>
<td>$50,000</td>
<td>3</td>
<td>37.50%</td>
<td>8</td>
<td>100.00%</td>
</tr>
</tbody>
</table>

Figure 4
Combined Log and List Output Two-Way Tables

Output Added:  
---------------
Name: List
Label: LIST Frequencies
Template: Base.Freq.List
Path: Freq.Table1.List
---------------

<table>
<thead>
<tr>
<th>Income</th>
<th>College</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Cumulative Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>$40,000</td>
<td>0</td>
<td>3</td>
<td>37.50</td>
<td>3</td>
<td>37.50</td>
</tr>
<tr>
<td>$40,000</td>
<td>1</td>
<td>2</td>
<td>25.00</td>
<td>5</td>
<td>62.50</td>
</tr>
<tr>
<td>$50,000</td>
<td>0</td>
<td>1</td>
<td>12.50</td>
<td>6</td>
<td>75.00</td>
</tr>
<tr>
<td>$50,000</td>
<td>1</td>
<td>2</td>
<td>25.00</td>
<td>8</td>
<td>100.00</td>
</tr>
</tbody>
</table>

To edit the template, we submit again the PROC TEMPLATE code used for one-way tables. The only difference is that now the outermost edit/end block specifies Base.Freq.List as the template to be edited:
ods path work.templat(update) sashelp.tmplstm(read);
proc template;
edit Base.Freq.OneWayList;
edit Percent;
  header="; Relative Frequency ;";
  format= pctfmt.;
  justify= on;
end;
edit CumPercent;
  header = ";Cumulative; Relative Frequency;";
  format= pctfmt.;
  justify= on;
end;
end;
run;

The modified output is displayed in Figure 5.

<table>
<thead>
<tr>
<th>Income</th>
<th>College</th>
<th>Frequency</th>
<th>Relative Frequency</th>
<th>Cumulative Frequency</th>
<th>Cumulative Relative Frequency</th>
</tr>
</thead>
<tbody>
<tr>
<td>$40,000</td>
<td>0</td>
<td>3</td>
<td>37.50%</td>
<td>3</td>
<td>37.50%</td>
</tr>
<tr>
<td>$40,000</td>
<td>1</td>
<td>2</td>
<td>25.00%</td>
<td>5</td>
<td>62.50%</td>
</tr>
<tr>
<td>$50,000</td>
<td>0</td>
<td>1</td>
<td>12.50%</td>
<td>6</td>
<td>75.00%</td>
</tr>
<tr>
<td>$50,000</td>
<td>1</td>
<td>2</td>
<td>25.00%</td>
<td>8</td>
<td>100.00%</td>
</tr>
</tbody>
</table>

**SUMMARY**

Editing SAS templates is a convenient way to change the display of PROC FREQ tabulations. In combination with PROC FORMAT, SAS users can change the format of frequencies and percents. The current version of SAS does not use a template to control the output of two-way tables and the only way to use PROC TEM-PLATE is to display two-way tables as lists.

**ACKNOWLEDGEMENTS**

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.

**CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the authors at:

Nan Liu  
Adheris Inc.  
One Van de Graaff Drive  
Burlington, MA 01803  
Email: nan.liu@adheris.com

Marcelo Coca-Perraillon  
Adheris Inc.  
One Van de Graaff Drive  
Burlington, MA 01803  
Email: marcelo.coca@adheris.com
APPENDIX A

```plaintext
link Base.Freq.OneWayFreqs to Base.Freq.OneWayList / notes = "One-Way Frequency table";
define table Base.Freq.OneWayList;
  notes "Parent for One-Way Frequency table and LIST table";
dynamic page needlines plabel varlabel lw varjust gluef gluep;
column Line FVariable FListVariable Variable Frequency TestFrequency Percent TestPercent
  CumFrequency CumPercent;
header h1;
translate _val_=._ into "";
define h1;
  text varlabel;
  space = 1;
  split = "";
  spill_margin;
  highlight;
end;
define Line;
  header = 'Line';
  format_ndec = 0;
  format_width = lw;
  just = c;
  style = RowHeader;
  id;
end;
define FVariable;
  just = varjust;
  parent = Base.Freq.FVariable;
end;
define FListVariable;
  just = varjust;
  parent = Base.Freq.FVariable;
  id = OFF;
end;
define Variable;
  parent = Base.Freq.Variable;
end;
define Frequency;
  glue = gluef;
  parent = Base.Freq.Frequency;
end;
define TestFrequency;
  header = "; Test Frequency;";
  glue = 4;
  parent = Base.Freq.Frequency;
end;
define Percent;
  glue = gluep;
  parent = Base.Freq.Percent;
end;
define TestPercent;
  header = "; Test Percent;";
  glue = 3;
  parent = Base.Freq.Percent;
end;
define CumFrequency;
  header = "; Cumulative Frequency;";
  glue = 4;
  parent = Base.Freq.Frequency;
end;
```

Posters NESUG 2006
```plaintext
APPENDIX B

define table Base.Freq.OneWayList;
notes "Parent for One-Way Frequency table and LIST table";
dynamic page needlines plabel varlabel lw varjust gluef gluep;
column Line FVariable FListVariable Variable Frequency TestFrequency Percent TestPercent CumFrequency CumPercent;
header h1;
translate _val_=._ into "";
declare h1;
text varlabel;
space = 1;
split = "";
spill_margin;
highlight;
end;
declare Line;
header = "Line";
format_ndec = 0;
format_width = lw;
just = c;
style = RowHeader;
id;
end;
declare FVariable;
just = varjust;
parent = Base.Freq.FVariable;
end;
declare FListVariable;
just = varjust;
parent = Base.Freq.FVariable;
id = OFF;
end;
declare Variable;
parent = Base.Freq.Variable;
end;
declare Frequency;
parent = Base.Freq.Frequency;
end;
declare TestFrequency;
header = " Test ;Frequency";
parent = Base.Freq.Frequency;
end;
declare Percent;
parent = Base.Freq.Percent;
end;
declare TestPercent;
```
header = "Test; Percent;";
glue = 3;
parent = Base.Freq.Percent;
end;
define CumFrequency;
    header = "Cumulative; Frequency;";
glue = 4;
    parent = Base.Freq.Frequency;
end;
define CumPercent;
    header = "Cumulative; Percent;";
    parent = Base.Freq.Percent;
end;
required_space = needlines;
print_headers = plabel;
newpage = page;
underline;
use_name;
end;